

9/6.2

Fonction LOCATE-INPUT

Le générateur de jeux d'aventures que nous étudierons dans un des prochains compléments utilise une fonction Assembleur bien utile appelée **LOCATE-INPUT**. Cette fonction positionne le curseur à un endroit quelconque sur l'écran (équivalent à la fonction Basic **LOCATE**), et acquiert une chaîne de caractères (équivalent à la fonction Basic **INPUT**).

Pour recréer la fonction **LOCATE**, nous allons utiliser plusieurs macros du firmware :

- **TXT CUR ENABLE** pour déclarer l'existence du curseur,
- **TXT CUR ON** pour faire apparaître le curseur sur l'écran,
- **TXT SET CUR** pour positionner le curseur sur l'écran.

Pour recréer la fonction **INPUT**, nous allons également utiliser plusieurs macros du firmware :


- **KM WAIT KEY** pour acquérir un caractère au clavier,
- **TXT WR CHAR** pour afficher un caractère sur l'écran.

Reportez-vous en Partie 4, chapitre 2.7, pages 5 et 12 pour avoir plus de détails sur ces macros.

Pour acquérir une suite de caractères avec la fonction **LOCATE-INPUT**, les étapes suivantes doivent être respectées :

- validation de l'affichage du curseur ;
- positionnement du curseur ;
- boucle sur la macro **KM WAIT KEY** jusqu'à ce que la touche **<Enter>** soit pressée ;
- affichage des caractères entrés sur l'écran grâce à la macro **TXT WR CHAR** ;
- stockage des caractères en mémoire en gérant un pointeur. Pour permettre une plus grande souplesse, la touche **DEL** est prise en compte. Elle permet d'effacer le caractère qui se trouve à gauche du curseur.

Ces actions sont résumées dans l'organigramme suivant :



Le listing du programme Assembleur est le suivant :

```

1      ;Fonction LOCATE+INPUT
2      ;-----
3      ORG 933FH
4      LOAD 933FH
5      ; Macros FIRMWARE
6      SETCUR: EQU 0BB75H      ;TXT SET CURSOR
7      TXTOUT: EQU 0BB5DH      ;TXT WR CHAR
8      WAITK: EQU 0BB18H      ;KM WAIT KEY
9      TCUREN: EQU 0BB7BH      ;TXT CUR ENABLE
10     TCURON: EQU 0BB81H      ;TXT CUR ON
11     TCUROF: EQU 0BB84H      ;TXT CUR OFF
12     ;
13     ;EQU Clavier
14     RET: EQU 13              ;RETurn
15     DEL: EQU 7FH            ;DElete
16     BLANC: EQU 32           ;Espace
17     CTRLH: EQU 8            ;CTRL+H
18     ;
19     ;Reservation de zones
20     SAVHL: DS 2              ;Sauvegarde reg. HL
21     SAV2: DS 2              ;2eme Sauv. reg. HL
22     SAVA: DS 1              ;Sauvegarde reg. A
23     MAX: DS 1               ;Long. max INPUT
24     ;-----
25     ;
26     ;Fonction LOCATE+INPUT avec
27     ;gestion de la touche DEL
28     ;
29     ;Entree: H =Colonne du LOCATE
30     ;      L =Ligne du LOCATE

```

```

31          ;          A =Nb Max de caracteres
32          ;          DE=Adresse de stockage
33          ;
34          ;Point d'entree:LOCINP
35          ;
36          LOCINP:    EQU $          ;LOCATE+INPUT
37 9345 324493        LD  (MAX),A      ;Lgr Max
38 9348 223F93        LD  (SAVHL),HL   ;Sauv Colonne/Ligne
39 934B CD7BBB        CALL TCUREN     ;Cursor Enable
40 934E CDB1BB        CALL TCURON     ;Cursor ON
41 9351 CD75BB        CALL SETCUR     ;LOCATE
42 9354 0600          LD  B,0          ;Init pointeur texte
43 9356 2A3F93        LD  HL,(SAVHL)  ;Restit. pos curseur
44          IO:       EQU $
45 9359 CD18BB        CALL WAITK      ;KM WAIT KEY
46 935C FE7F          CP  DEL          ;Est-ce un DEL ?
47 935E 2826          JR  Z,I1         ;Oui
48 9360 FE0D          CP  RET          ;Est-ce un RET ?
49 9362 284F          JR  Z,I3         ;Oui
50 9364 FE08          CP  CTRLH       ;Est-ce un CTRL+H ?
51 9366 2855          JR  Z,I5         ;Oui
52 9368 324393        LD  (SAVA),A    ;Ni DEL, ni RET
53          III:     EQU $
54 936B 3A4493        LD  A,(MAX)
55 936E B8            CP  B
56 936F 38EB          JR  C,I0         ;Pas possible d'ecrire
57 9371 C5            PUSH BC
58 9372 D5            PUSH DE
59 9373 E5            PUSH HL
60 9374 3A4393        LD  A,(SAVA)
61 9377 CD5DBB        CALL TXTOUT     ;Ecriture
62 937A E1            POP  HL

```

```

63 937B D1          POP  DE
64 937C C1          POP  BC
65 937D 04          INC  B
66 937E 24          INC  H           ;Pos curseur + 1
67 937F 3A4393      LD   A,(SAVA)      ;Caractere tape
68 9382 12          LD   (DE),A       ;Memorisation
69 9383 13          INC  DE           ;Memoire suivante
70 9384 18D3        JR   IO           ;Boucle de saisie
71                I1: EQU  $           ;Traitement du DEL
72 9386 224193      LD   (SAV2),HL     ;Sauv pos curseur
73 9389 D5          PUSH DE
74 938A EB          EX   DE,HL
75 938B 2A3F93      LD   HL,(SAVHL)
76 938E 7A          LD   A,D
77 938F BC          CP   H
78 9390 2828        JR   Z,I4         ;DEL Impossible
79 9392 D1          POP  DE           ;DEL Possible
80 9393 05          DEC  B
81 9394 2A4193      LD   HL,(SAV2)    ;Pos curseur
82 9397 25          DEC  H
83 9398 224193      LD   (SAV2),HL
84 939B CD75BB      CALL SETCUR       ;LOCATE
85 939E C5          PUSH BC
86 939F D5          PUSH DE
87 93A0 3E20        LD   A,BLANC      ;Code "Blanc"
88 93A2 CD5DBB      CALL TXTOUT       ;Ecriture
89 93A5 D1          POP  DE
90 93A6 C1          POP  BC
91 93A7 2A4193      LD   HL,(SAV2)    ;Restit @ Curseur
92 93AA CD75BB      CALL SETCUR
93 93AD 2A4193      LD   HL,(SAV2)    ;Restit @ Curseur
94 93B0 1B          DEC  DE           ;Pointeur memoire - 1

```

```

95 93B1 18A6          JR   IO          ;Boucle de saisie
96                   I3:   EQU   $          ;Traitement RET
97 93B3 3EFF          LD   A,OFFH      ;Terminateur
98 93B5 12            LD   (DE),A      ;Ecriture memoire
99 93B6 CD84BB        CALL TCURDF      ;Text Cursor Off
100 93B9 C9           RET              ;Retour au prog. appelant
101                   I4:   EQU   $          ;DEL Impossible
102 93BA D1           POP   DE
103 93BB 189C         JR   IO          ;Retour saisie
104                   I5:   EQU   $
105 93BD 324393       LD   (SAVA),A
106 93C0 3ACC93       LD   A,(HELP)
107 93C3 B7           OR   A
108 93C4 28A5         JR   Z,III       ;Fausse alerte
109 93C6 3E01         LD   A,1
110 93C8 32CD93       LD   (CTHACT),A  ;CTRL+H Active
111 93CB C9           RET
112                   ;-----
113                   HELP:  DS   1          ;=1 Si fn HELP active
114                   CTHACT: DS   1          ;=1 Si CTRL+H active
115                   ;-----
116                   END

```

- Lignes 1 à 24 : Déclarations diverses de constantes et de zones mémoires.
- Lignes 39 et 40 : Validation et affichage du curseur.
- Ligne 41 : Positionnement du curseur.
- Ligne 45 : Attente d'un caractère au clavier.
- Ligne 61 : Affichage du caractère tapé.
- Ligne 68 : Ecriture en mémoire du caractère tapé.
- Lignes 79 à 95 : Traitement de la touche DEL.

Pour activer la fonction **LOCATE-INPUT**, le petit programme suivant sera utilisé. Ce programme permet également d'analyser la cohérence de la chaîne entrée en fonction de la salle courante.

```

1           ;Activation de INPUT et
2           ;analyse de la reponse utilisateur
3           ;-----
4           ORG 91A0H
5           LOAD 91A0H
6           SABAS: DS 1           ;No Salle BASIC
7           SALLE: EQU 907FH      ;Salle ds ANALYS
8           INPUT: EQU 9345H      ;Input ASM
9           ANALYS: EQU 908DH     ;Anal. syntaxiq.
10          ;-----
11 91A1 211701 LD HL,117H        ;Colonne/Ligne
12 91A4 3E26 LD A,38             ;Nbre max caracteres
13 91A6 110043 LD DE,4300H      ;@ Stockage reponse
14 91A9 CD4593 CALL INPUT
15 91AC 3AA091 LD A,(SABAS)      ;Salle BASIC
16 91AF 327F90 LD (SALLE),A
17 91B2 CD8D90 CALL ANALYS      ;Analyse reponse
18 91B5 C9 RET                   ;Retour au BASIC
19          END

```

- Ligne 11 : Positionnement de la colonne et de la ligne de début de saisie dans le registre HL.
- Ligne 12 : Taille maximale de la chaîne saisie dans le registre A.
- Ligne 13 : Adresse de stockage de la réponse dans le registre DE.
- Ligne 14 : **LOCATE-INPUT**.
- Ligne 17 : Analyse syntaxique de la chaîne entrée.

